

power rsquared — Power analysis for an R^2 test in a multiple linear regression

Description	Quick start	Menu	Syntax
Options	Remarks and examples	Stored results	Methods and formulas
Reference	Also see		

Description

`power rsquared` computes sample size, power, or target R^2 for an R^2 test in a multiple linear regression. An R^2 test is an F test for the coefficient of determination, R^2 , which is used to test the significance of all coefficients or of a subset of coefficients in a regression model.

By default, `power rsquared` computes sample size for a test of all coefficients given power and the R^2 of the tested model, R_T^2 . Instead of the sample size, it can compute power given sample size and R_T^2 or the target R_T^2 given sample size and power.

If the number of control covariates is provided, `power rsquared` computes sample size for a test of a subset of coefficients given power, the R^2 of the full model, R_F^2 , and the R^2 of the reduced model, R_R^2 . It can also compute power given sample size, R_R^2 , and R_F^2 or the target R_F^2 given sample size, power, and R_R^2 .

See [\[PSS-2\] power](#) for a general introduction to the `power` command using hypothesis tests.

Quick start

Testing all coefficients

Sample size for a test of $H_0: R_T^2 = 0$ versus $H_a: R_T^2 \neq 0$ given alternative R_T^2 of 0.10 and 2 tested covariates using default power 0.8 and significance level $\alpha = 0.05$

```
power rsquared 0.10, ntested(2)
```

Also use values of R_T^2 equal to 0.11, 0.12, 0.13, and 0.14, and display results in a table

```
power rsquared (0.10(0.01)0.14), ntested(2)
```

As above, but display results in a graph of sample size versus R_T^2

```
power rsquared (0.10(0.01)0.14), ntested(2) graph
```

Power for a sample size of 80

```
power rsquared 0.10, ntested(2) n(80)
```

Effect size and target R_T^2 for a sample size of 80 with power 0.9

```
power rsquared, ntested(2) n(80) power(0.9)
```

Testing a subset of coefficients

Sample size for a test of $H_0: R_F^2 = R_R^2$ versus $H_a: R_F^2 \neq R_R^2$ given R^2 of the reduced model of 0.10, the hypothesized R^2 of the full model of 0.15, 2 tested covariates, and 3 control covariates using default power 0.8 and significance level $\alpha = 0.05$

```
power rsquared 0.10 0.15, ntested(2) ncontrol(3)
```

Also use values of R_F^2 equal to 0.11, 0.12, 0.13, 0.14, and 0.15, and display results in a table

```
power rsquared 0.10 (0.11(0.01)0.15), ntested(2) ncontrol(3)
```

As above, but display results in a graph of sample size versus R_F^2

```
power rsquared 0.10 (0.11(0.01)0.15), ntested(2) ncontrol(3) graph
```

Power for a sample size of 80

```
power rsquared 0.10 0.15, ntested(2) ncontrol(3) n(80)
```

Effect size and target R^2 for a sample size of 80 with power of 0.9

```
power rsquared 0.10, ntested(2) ncontrol(3) n(80) power(0.9)
```

Menu

Statistics > Power, precision, and sample size

Syntax

Compute sample size

Test all coefficients

```
power rsquared  $R_T^2$  [ , power(numlist) options ]
```

Test a subset of coefficients

```
power rsquared  $R_R^2$   $R_F^2$  , ncontrol(numlist) [power(numlist) options ]
```

Compute power

Test all coefficients

```
power rsquared  $R_T^2$  , n(numlist) [options ]
```

Test a subset of coefficients

```
power rsquared  $R_R^2$   $R_F^2$  , ncontrol(numlist) n(numlist) [options ]
```

Compute effect size and target R^2

Test all coefficients

```
power rsquared , n(numlist) power(numlist) [options ]
```

Test a subset of coefficients

```
power rsquared  $R_R^2$  , ncontrol(numlist) n(numlist) power(numlist) [options ]
```

where R_T^2 is the hypothesized R^2 of the tested model under the alternative hypothesis when testing all coefficients in the model, R_R^2 is the R^2 of the reduced model, and R_F^2 is the hypothesized R^2 of the full model when testing a subset of coefficients in the model.

R_T^2 , R_R^2 , and R_F^2 may each be specified either as one number or as a list of values in parentheses (see [U] 11.1.8 numlist).

<i>options</i>	Description
Main	
* <u>a</u> lpha(<i>numlist</i>)	significance level; default is alpha(0.05)
* <u>p</u> ower(<i>numlist</i>)	power; default is power(0.8)
* <u>b</u> eta(<i>numlist</i>)	probability of type II error; default is beta(0.2)
* <u>n</u> (<i>numlist</i>)	sample size; required to compute power or effect size
<u>n</u> fractional	allow fractional sample size
* <u>n</u> tested(<i>numlist</i>)	number of tested covariates
* <u>n</u> control(<i>numlist</i>)	number of control covariates; required for testing a subset of coefficients
* <u>d</u> iff(<i>numlist</i>)	difference between the R^2 of the full and the reduced model, $R_F^2 - R_R^2$; specify instead of the R^2 of the full model, R_F^2 , when testing a subset of coefficients
<u>p</u> arallel	treat number lists in starred options as parallel when multiple values per option are specified (do not enumerate all possible combinations of values)
Table	
[<u>n</u> o] <u>t</u> able [(<i>tablespec</i>)]	suppress table or display results as a table; see [PSS-2] power, table
<u>s</u> aving(<i>filename</i> [, replace])	save the table data to <i>filename</i> ; use <i>replace</i> to overwrite existing <i>filename</i>
Graph	
<u>g</u> raph [(<i>graphopts</i>)]	graph results; see [PSS-2] power, graph
Iteration	
<u>i</u> nit(#)	initial value for sample size or R^2 of tested model in the case of testing all coefficients and R^2 difference in the case of testing a subset of coefficients
<u>i</u> terate(#)	maximum number of iterations; default is <code>iterate(500)</code>
<u>t</u> olerance(#)	parameter tolerance; default is <code>tolerance(1e-12)</code>
<u>f</u> tolerance(#)	function tolerance; default is <code>ftolerance(1e-12)</code>
[<u>n</u> o] <u>l</u> og	suppress or display iteration log
[<u>n</u> o] <u>d</u> ots	suppress or display iterations as dots
<u>n</u> otitle	suppress the title

*Specifying a list of values in at least two starred options, or at least two command arguments, or at least one starred option and one argument results in computations for all possible combinations of the values; see [U] 11.1.8 **numlist**. Also see the `parallel` option.

`collect` is allowed; see [U] 11.1.10 **Prefix commands**.

`notitle` does not appear in the dialog box.

where *tablespec* is

column [:*label*] [*column* [:*label*] [...]] [, *tableopts*]

column is one of the columns defined [below](#), and *label* is a column label (may contain quotes and compound quotes).

<i>column</i>	Description	Symbol
<code>alpha</code>	significance level	α
<code>power</code>	power	$1 - \beta$
<code>beta</code>	type II error probability	β
<code>N</code>	number of subjects	N
<code>delta</code>	effect size	δ
<code>R2_T</code>	R^2 of the tested model	R_T^2
<code>R2_R</code>	R^2 of the reduced model	R_R^2
<code>R2_F</code>	R^2 of the full model	R_F^2
<code>R2_diff</code>	difference of R^2 between full and reduced models	R_D^2
<code>ntested</code>	number of tested covariates	N_T
<code>ncontrol</code>	number of control covariates	N_C
<code>target</code>	target parameter; synonym for <code>R2_T</code> or <code>R2_diff</code>	
<code>_all</code>	display all supported columns	

Column `beta` is shown in the default table in place of column `power` if specified.

Column `R2_T` is shown in the default table for a test of all coefficients and is not available if `ncontrol()` is specified.

Columns `R2_R`, `R2_F`, `R2_diff`, and `ncontrol` are shown in the default table for a test of a subset of coefficients and only available if `ncontrol()` is specified.

For a test of all coefficients, `target` is `R2_T`. For a test of a subset of coefficients, `target` is `R2_diff`.

Options

Main

`alpha()`, `power()`, `beta()`, `n()`, `nfractional`; see [PSS-2] **power**. The `nfractional` option is allowed only for sample-size determination.

`ntested(numlist)` specifies the number of tested covariates. The default is `ntested(1)`.

`ncontrol(numlist)` specifies the number of control covariates or the number of the covariates in the reduced model. This option is required for testing a subset of coefficients. If the option is not specified, all coefficients are assumed to be tested.

`diff(numlist)` specifies the difference between the R^2 of the full and reduced models, $R_F^2 - R_R^2$, when computing sample size or power for a test of a subset of coefficients. You may specify either the R^2 of the full model, R_F^2 , as a command argument or the difference $R_F^2 - R_R^2$ in the `diff()` option. This option is not allowed with effect-size computation.

`parallel`; see [PSS-2] **power**.

Table

`table`, `table()`, `notable`; see [PSS-2] **power, table**.

`saving()`; see [PSS-2] **power**.

Graph

`graph`, `graph()`; see [PSS-2] **power, graph**. Also see the *column* table for a list of symbols used by the graphs.

Iteration

`init(#)` specifies the initial value of the sample size for the sample-size determination or the initial value of the R^2 of the tested model in the case of testing all coefficients and the difference between the R^2 of the full and reduced models in the case of testing a subset of coefficients for the effect-size determination. The default is to use a bisection search method to compute an initial value.

`iterate()`, `tolerance()`, `ftolerance()`, `log`, `nolog`, `dots`, `nodots`; see [PSS-2] **power**.

The following option is available with `power rsquared` but is not shown in the dialog box: `notitle`; see [PSS-2] **power**.

Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

Introduction

Using power rsquared

Computing sample size

Computing power

Computing effect size and target R^2

Performing hypothesis tests on the coefficients

`power rsquared` computes sample size, power, and the target R^2 for a multiple linear regression R^2 test. See [PSS-2] **Intro (power)** for a general introduction to power and sample-size analysis, and see [PSS-2] **power** for a general introduction to the `power` command using hypothesis tests.

Introduction

In contrast to a simple linear regression, a multiple regression framework allows researchers to control for additional variables that may better predict or explain the variation in the dependent variable of interest. The fit of the model, as measured by the R^2 statistic, sheds light on the efficacy of the multiple regression model in explaining the variation of the dependent variable.

Several scenarios arise for testing the fit of a multiple regression model. Consider an example where a researcher is interested in the effect of gender and education on wage,

$$y_{\text{wage}} = \beta_0 + \beta_{\text{edu}}x_{\text{edu}} + \beta_{\text{gender}}x_{\text{gender}} + \varepsilon \quad (1)$$

where the error term ε is independently and normally distributed with mean zero and constant standard deviation σ .

An F test may be performed for testing the joint significance of the coefficients on education and gender. The null hypothesis may be stated as $H_0: \beta_{\text{educ}} = \beta_{\text{gender}} = 0$; x_{edu} and x_{gender} are the tested covariates. This is equivalent to the null hypothesis $H_0: R_T^2 = 0$, where R_T^2 is the coefficient of determination, or alternatively, the variation of the dependent variable explained by the tested model.

Alternatively, a researcher may be interested in whether experience x_{exp} and location x_{loc} add further information in explaining wage variation after controlling for x_{edu} and x_{gender} :

$$y_{\text{wage}} = \beta_0 + \beta_{\text{exp}}x_{\text{exp}} + \beta_{\text{loc}}x_{\text{loc}} + \beta_{\text{edu}}x_{\text{edu}} + \beta_{\text{gender}}x_{\text{gender}} + \varepsilon \quad (2)$$

An F test may also be performed for testing the joint significance of the coefficients on experience and location. The null hypothesis may be stated as $H_0: \beta_{\text{exp}} = \beta_{\text{loc}} = 0$; x_{exp} and x_{loc} are now the tested covariates, and x_{edu} and x_{gender} are the control covariates. An equivalent test can be constructed based on the R^2 . The null hypothesis is then $H_0: R_F^2 = R_R^2$, where R_F^2 is the R^2 of the full model (2) and R_R^2 is the R^2 of the reduced model (1).

The `power rsquared` command provides power and sample-size analysis for the test of R^2 using an F test. For power analysis for a partial-correlation test in a multiple linear regression, see [PSS-2] [power pcorr](#). For power analysis for a slope test in a simple linear regression, see [PSS-2] [power oneslope](#).

Using power rsquared

`power rsquared` computes sample size, power, or target R^2 for an R^2 test in a multiple linear regression. By default, all computations are performed at the significance level of 0.05. You may change the significance level by specifying the `alpha()` option.

By default, the number of tested covariates is set to 1. You can change the number of tested covariates with the `ntested()` option. All computations assume that the model includes a constant. To test a subset of coefficients, you must also specify the `ncontrol()` option.

To compute sample size for testing all coefficients in the model, you must specify the R^2 of the tested model, R_T^2 . For testing a subset of coefficients, you must specify the R^2 of the reduced model, R_R^2 , and the R^2 of the full model, R_F^2 . For either test, you can specify the power of the test in the `power()` option. The default power is set to 0.8.

To compute power, you must specify the sample size in the `n()` option. To test all coefficients in the model, you must also specify R_T^2 . To test a subset of coefficients, you must also specify R_R^2 and R_F^2 .

When computing sample size or power for a subset of coefficients, you can specify the difference between the R^2 of the full and reduced models in the `diff()` option instead of R_F^2 .

To compute effect size, which is defined as the ratio of R^2 explained by the tested covariates to the variance explained by the model error, you must specify the sample size in the `n()` option and the power of the test in the `power()` option. For a test of all coefficients, `power rsquared` reports the effect size and R_T^2 . For a test of a subset of coefficients, you must also specify R_R^2 to obtain the effect size. For this test, `power rsquared` reports the effect size and the difference between the R^2 statistics of the full and reduced models.

By default, the computed sample size is rounded up. You can specify the `nfractional` option to see the corresponding fractional sample size; see [Fractional sample sizes](#) in [PSS-4] [Unbalanced designs](#) for an example. The `nfractional` option is allowed only for sample-size determination.

`power rsquared`'s computations of sample size and effect size require iteration because the denominator degrees of freedom of the noncentral F distribution depends on the sample size, and the noncentrality parameter depends on the sample size and effect size. The default initial values are obtained using a bisection search method. You can use the `init()` option to specify your own value. The initial value of the sample size must be greater than the number of parameters in the multiple regression model. See [PSS-2] [power](#) for the descriptions of other options that control the iteration procedure.

In the following sections, we describe the use of `power rsquared` accompanied by examples for computing sample size, power, and target R^2 .

Computing sample size

To compute sample size for testing all coefficients in the model, you must specify the R^2 of the tested model, R^2_T . For testing a subset of coefficients, you must specify the R^2 of the reduced model, R^2_R , the R^2 of the full model, R^2_F , and the number of control covariates in `ncontrol()`. For either test, a default power of 0.8 is assumed if `power()` is not specified, and one tested covariate is assumed if `ntested()` is not specified.

► Example 1: Sample size for testing all coefficients

Consider an example from [Cohen \(1988, 424\)](#) where a psychologist investigates a selection procedure based on job candidates' demographic characteristics used to predict success in a sales position. The five variables are age, education, prior experience, verbal aptitude, and extraversion.

Previous studies found that the addition of all five variables accounted for 10% of the variance in the dependent variable, that is to say, an R^2 of 0.1. We are designing a new study and want to determine the required sample size for detecting this previously observed R^2 so that we achieve 80% power at a 5% significance level. We do this by typing

```
. power rsquared 0.1, ntested(5)
Performing iteration ...
Estimated sample size for multiple linear regression
F test for R2 testing all coefficients
H0: R2_T = 0 versus Ha: R2_T != 0
Study parameters:
      alpha =    0.0500
      power =    0.8000
      delta =    0.1111
      R2_T   =    0.1000
      ntested =      5
Estimated sample size:
      N =    122
```

We find that a sample of 122 subjects is required to detect an R^2 of 0.1 with 80% power using a 5%-level test. The effect size (`delta`) is calculated using the given R^2 of the model; see [Methods and formulas](#) for details.

As we mentioned in [Using power rsquared](#), sample-size computation requires iteration. The iteration log is suppressed by default, but you can display it by specifying the `log` option.



► Example 2: Sample size for testing a subset of coefficients

Continuing with [example 1](#), suppose that data for three of the variables—age, education, and prior experience—are readily available to the investigator. However, data for the other two variables—verbal aptitude and extraversion—are costly to obtain.

From previous studies, age, education, and prior experience explain about 6% of the variance in the dependent variable. The decision to include verbal aptitude and extraversion is deemed important only if their addition explains an additional 4% of the variance in the dependent variable; together the five variables should explain about 10% of the variance in the dependent variable. We will construct a test-case study to see the minimum sample size required to detect a 4% change in the R^2 . We compute the minimum sample size required with 80% power at a 5% significance level:


```
. power rsquared 0.06 0.1, ntested(2) ncontrol(3)
Performing iteration ...
Estimated sample size for multiple linear regression
F test for R2 testing subset of coefficients
H0: R2_F = R2_R versus Ha: R2_F != R2_R
Study parameters:
    alpha =    0.0500
    power =    0.8000
    delta =    0.0444
    R2_R =    0.0600
    R2_F =    0.1000
    R2_diff = 0.0400
    ncontrol =      3
    ntested =      2
Estimated sample size:
    N =      220
```

We find that a sample of 220 subjects is required to detect an increase in R^2 by 4% (`R2_diff`) after adding verbal aptitude and extraversion with 80% power using a 5% level test.

◀

► Example 3: Specifying difference between the R^2 of the full and reduced models

Instead of using the R^2 of the full model 0.1 as in [example 2](#), we can specify the difference between the R^2 of the full and reduced models, 0.04, in the `diff()` option.

```
. power rsquared 0.06, ntested(2) ncontrol(3) diff(0.04)
Performing iteration ...
Estimated sample size for multiple linear regression
F test for R2 testing subset of coefficients
H0: R2_F = R2_R versus Ha: R2_F != R2_R
Study parameters:
    alpha =    0.0500
    power =    0.8000
    delta =    0.0444
    R2_R =    0.0600
    R2_F =    0.1000
    R2_diff = 0.0400
    ncontrol =      3
    ntested =      2
Estimated sample size:
    N =      220
```

We obtain the same results as in [example 2](#).

◀

Computing power

To compute power, you must specify the sample size in the `n()` option. To test all coefficients in the model, you must also specify the R^2 of the tested model, R_T^2 . To test a subset of coefficients, you must also specify the R^2 of the reduced, R_R^2 , and full models, R_F^2 , and the number of control covariates in `ncontrol()`. The number of tested covariates is assumed to be one if `ntested()` is not specified.

▷ Example 4: Power for testing all coefficients in a multiple regression

Continuing with [example 1](#), suppose that we are designing a new study and anticipate obtaining a sample of 100 subjects. To compute the power corresponding to this sample size given the study parameters from example 1, we specify the sample size of 100 in the `n()` option:

```
. power rsquared 0.1, ntested(5) n(100)
Estimated power for multiple linear regression
F test for R2 testing all coefficients
H0: R2_T = 0 versus Ha: R2_T != 0
Study parameters:
    alpha =    0.0500
      N =      100
    delta =    0.1111
    R2_T =    0.1000
    ntested =      5
Estimated power:
    power =    0.7014
```

For the smaller sample size, we achieve a lower power of about 70%.

◀

▷ Example 5: Power for testing a subset of coefficients

Continuing with [example 2](#), suppose that we are designing a new study and anticipate a sample of 200 subjects. To compute the power corresponding to this sample size given the study parameters from example 2, we specify the sample size of 200 in the `n()` option:

```
. power rsquared 0.06 0.1, ntested(2) ncontrol(3) n(200)
Estimated power for multiple linear regression
F test for R2 testing subset of coefficients
H0: R2_F = R2_R versus Ha: R2_F != R2_R
Study parameters:
    alpha =    0.0500
      N =      200
    delta =    0.0444
    R2_R =    0.0600
    R2_F =    0.1000
    R2_diff =  0.0400
    ncontrol =    3
    ntested =    2
Estimated power:
    power =    0.7583
```

With a smaller sample size, the power of the test decreases to about 76%.

◀

▷ Example 6: Multiple values of study parameters

Continuing with [example 5](#), we want to investigate the effect of sample size on power, so we specify a list of sample sizes in the `n()` option:

```
. power rsquared 0.06 0.1, n(50 100 200 400 800) ntested(2) ncontrol(3)
Estimated power for multiple linear regression
F test for R2 testing subset of coefficients
H0: R2_F = R2_R versus Ha: R2_F != R2_R
```

alpha	power	N	delta	R2_R	R2_F	R2_diff	ntested	ncontrol
.05	.2328	50	.04444	.06	.1	.04	2	3
.05	.4431	100	.04444	.06	.1	.04	2	3
.05	.7583	200	.04444	.06	.1	.04	2	3
.05	.9719	400	.04444	.06	.1	.04	2	3
.05	.9999	800	.04444	.06	.1	.04	2	3

As expected, when the sample size increases, the power tends to get closer to 1.

For multiple values of parameters, the results are automatically displayed in a table, as we see above. For more examples of tables, see [PSS-2] [power, table](#). If you wish to produce a power plot, see [PSS-2] [power, graph](#).

◀

Computing effect size and target R^2

For a test of all coefficients, effect size δ is defined as the ratio of the R^2 of the tested model to the variance explained by the model error term, $\delta = R_T^2 / (1 - R_T^2)$. For a test of a subset of coefficients, effect size δ is defined as the ratio of the difference between the R^2 of the full and reduced models to the variance explained by the model error term, $\delta = (R_F^2 - R_R^2) / (1 - R_F^2)$.

Sometimes, we may be interested in determining the minimum detectable effect and the corresponding target R^2 that yield a statistically significant result for a prespecified sample size and power. In this case, we must specify power and sample size. To test a subset of coefficients, we must also specify the R^2 of the reduced model and the number of control covariates.

▶ Example 7: Minimum detectable value for the R^2 of the tested model

Continuing with [example 4](#), we may also be interested in finding the minimum value for the R^2 of the tested model that can be detected with a power of 80% given a sample of 100 subjects. To compute this, we specify the sample size of 100 in the `n()` option and power of 0.8 in the `power()` option. As in [example 4](#), we use the same value of 5 for the number of tested covariates.

```
. power rsquared, n(100) power(0.8) ntested(5)
Performing iteration ...
Estimated R-squared for multiple linear regression
F test for R2 testing all coefficients
H0: R2_T = 0 versus Ha: R2_T != 0
Study parameters:
    alpha =    0.0500
    power =    0.8000
      N =      100
    ntested =    5
Estimated effect size and R-squared:
    delta =    0.1360
    R2_T =    0.1197
```

The minimum detectable value for the R^2 of the tested model is 0.1197, which corresponds to the effect size of 0.1360. Compared with [example 4](#), we would detect a slightly larger value for the R^2 of the tested model.

◀

▷ Example 8: Minimum detectable value of the R^2 difference

Continuing with [example 5](#), we may also be interested in finding the minimum difference between the R^2 of the full and reduced models that can be detected with a power of 80% given a sample of 200 subjects. To compute this, we specify the R^2 of the reduced model of 0.06 as the command argument and also specify the sample size of 200 in the `n()` option and a power of 0.8 in the `power()` option. As in [example 5](#), we use 2 tested covariates and 3 control covariates.

```
. power rsquared 0.06, n(200) power(0.8) ntested(2) ncontrol(3)
Performing iteration ...
Estimated R-squared for multiple linear regression
F test for R2 testing subset of coefficients
H0: R2_F = R2_R versus Ha: R2_F != R2_R
Study parameters:
      alpha =    0.0500
      power =    0.8000
         N =     200
      R2_R =    0.0600
ncontrol =     3
ntested =     2
Estimated effect size and R-squared:
      delta =    0.0489
R2_diff =    0.0438
      R2_F =    0.1038
```

The minimum detectable value for the R^2 difference is 0.0438, which corresponds to an effect size of 0.0489. In [example 2](#), we assumed the same power and significance levels, and we found that we need 220 subjects to detect an R^2 difference of 0.04. With the smaller sample of 200, the minimum detectable R^2 difference is slightly larger.

◀

Performing hypothesis tests on the coefficients

Suppose we wish to test the hypothesis that, controlling for other variables, a group of variables has no effect in explaining the variance of the dependent variable. We can use the `regress` command to estimate the coefficients and the `test` command to perform a hypothesis test.

▷ Example 9: Joint test of the coefficients

Consider `auto.dta`, which contains various characteristics of 74 cars. Suppose that our study goal is to investigate whether headroom (`headroom`) and trunk space (`trunk`) have any effect on the price of cars after controlling for their weight (`weight`) and mileage (`mpg`). We can fit the full linear model and obtain the R^2 using `regress`.

```
. use https://www.stata-press.com/data/r17/auto
(1978 automobile data)
```

```
. regress price headroom trunk weight mpg
```

Source	SS	df	MS	Number of obs	=	74
Model	204838391	4	51209597.9	F(4, 69)	=	8.21
Residual	430227005	69	6235173.98	Prob > F	=	0.0000
				R-squared	=	0.3225
				Adj R-squared	=	0.2833
Total	635065396	73	8699525.97	Root MSE	=	2497

price	Coefficient	Std. err.	t	P> t	[95% conf. interval]
headroom	-726.5434	462.0322	-1.57	0.120	-1648.272 195.1856
trunk	23.04248	108.3649	0.21	0.832	-193.1396 239.2246
weight	2.011936	.7036432	2.86	0.006	.6082062 3.415666
mpg	-54.79153	85.91635	-0.64	0.526	-226.19 116.6069
_cons	3114.94	3648.08	0.85	0.396	-4162.779 10392.66

The R^2 of the full model is around 0.32. We can test the joint hypothesis $H_0: \beta_{\text{headroom}} = \beta_{\text{trunk}} = 0$ using the `test` command.

```
. test headroom trunk
( 1) headroom = 0
( 2) trunk = 0
      F( 2, 69) = 1.48
      Prob > F = 0.2337
```

We fail to reject the null hypothesis at the 5% significance level; the p -value > 0.05 .

Suppose we wish to design a new similar study. We use the estimates from this study to perform a sample-size analysis. First, we need to fit the reduced model to obtain the estimate of its R^2 .

```
. regress price weight mpg
```

Source	SS	df	MS	Number of obs	=	74
Model	186321280	2	93160639.9	F(2, 71)	=	14.74
Residual	448744116	71	6320339.67	Prob > F	=	0.0000
				R-squared	=	0.2934
				Adj R-squared	=	0.2735
Total	635065396	73	8699525.97	Root MSE	=	2514

price	Coefficient	Std. err.	t	P> t	[95% conf. interval]
weight	1.746559	.6413538	2.72	0.008	.467736 3.025382
mpg	-49.51222	86.15604	-0.57	0.567	-221.3025 122.278
_cons	1946.069	3597.05	0.54	0.590	-5226.245 9118.382

The R^2 of the reduced model is around 0.29. Next, we specify this number as well as the R^2 of the full model, 0.32, with `power rsquared` to perform a sample-size analysis.

```

. power rsquared 0.29 0.32, ntested(2) ncontrol(2)
Performing iteration ...
Estimated sample size for multiple linear regression
F test for R2 testing subset of coefficients
H0: R2_F = R2_R versus Ha: R2_F != R2_R
Study parameters:
      alpha =    0.0500
      power =    0.8000
      delta =    0.0441
      R2_R   =    0.2900
      R2_F   =    0.3200
      R2_diff =    0.0300
      ncontrol =      2
      ntested =      2
Estimated sample size:
      N =      222

```

We find that a sample size of 222 is required to detect an increase in R^2 by 0.03 with 80% power using a 5%-level two-sided test.

4

Stored results

`power rsquared` stores the following in `r()`:

Scalars

<code>r(alpha)</code>	significance level
<code>r(power)</code>	power
<code>r(beta)</code>	probability of a type II error
<code>r(delta)</code>	effect size
<code>r(N)</code>	sample size
<code>r(nfractional)</code>	1 if <code>nfractional</code> is specified, 0 otherwise
<code>r(R2_T)</code>	R^2 of the tested model
<code>r(R2_R)</code>	R^2 of the reduced model
<code>r(R2_F)</code>	R^2 of the full model
<code>r(R2_diff)</code>	difference between R^2 of the full and reduced models
<code>r(ntested)</code>	number of tested covariates
<code>r(ncontrol)</code>	number of control covariates
<code>r(separator)</code>	number of lines between separator lines in the table
<code>r(divider)</code>	1 if <code>divider</code> is requested in the table, 0 otherwise
<code>r(init)</code>	initial value for sample size or for R^2
<code>r(maxiter)</code>	maximum number of iterations
<code>r(iter)</code>	number of iterations performed
<code>r(tolerance)</code>	requested parameter tolerance
<code>r(deltax)</code>	final parameter tolerance achieved
<code>r(ftolerance)</code>	requested distance of the objective function from zero
<code>r(function)</code>	final distance of the objective function from zero
<code>r(converged)</code>	1 if iteration algorithm converged, 0 otherwise

Macros

<code>r(type)</code>	test
<code>r(method)</code>	<code>rsquared</code>
<code>r(columns)</code>	displayed table columns
<code>r(labels)</code>	table column labels
<code>r(widths)</code>	table column widths
<code>r(formats)</code>	table column formats

Matrices

<code>r(pss_table)</code>	table of results
---------------------------	------------------

Methods and formulas

Methods and formulas are presented under the following headings:

Introduction

Testing all coefficients

Testing a subset of coefficients: R^2 of full versus reduced models

Testing a subset of coefficients: Partial multiple correlation

Introduction

This section subsumes the *Methods and formulas* for [PSS-2] **power pcorr**.

Consider a multiple linear regression model of a dependent variable y_i on $k + p$ fixed covariates x_i in a sample of n subjects,

$$y_i = \beta_0 + \beta_1 x_{1i} + \beta_2 x_{2i} + \cdots + \beta_k x_{ki} + \beta_{k+1} x_{(k+1)i} + \cdots + \beta_{k+p} x_{(k+p)i} + \varepsilon_i$$

where ε_i 's are independently and normally distributed with mean zero and constant standard deviation σ . Suppose that we are interested in testing the significance of the coefficients in the model and we construct an F statistic using the effect size δ ; see subsections below for the definition of the effect size specific to each test. Under the alternative hypothesis, the F statistic

$$F = \frac{\delta/\nu_1}{1/\nu_2}$$

follows a noncentral F distribution with noncentrality parameter $\lambda = n\delta$ and ν_1 numerator and ν_2 denominator degrees of freedom. The formulas of the F test are based on [Cohen \(1988\)](#).

Let α be the significance level, β be the probability of a type II error, and $F_{\nu_1, \nu_2, \alpha}$ denote the α th quantile of an F distribution with ν_1 numerator and ν_2 denominator degrees of freedom.

The power $\pi = 1 - \beta$ is computed using

$$\pi = 1 - F_{\nu_1, \nu_2, \lambda}(F_{\nu_1, \nu_2, 1-\alpha}) \quad (3)$$

where $F_{\nu_1, \nu_2, \lambda}(\cdot)$ is the cumulative noncentral F distribution with a noncentrality parameter λ .

Sample size and effect size are obtained by iteratively solving the nonlinear equations in (3) for n and δ . If the `nfractional` option is not specified, the computed sample size is rounded up.

Details of the effect size δ and the F statistics are given in the following sections.

Testing all coefficients

Consider first the hypothesis that all the coefficients are zero, $H_0: \beta_1 = \beta_2 = \cdots = \beta_{k+p} = 0$. Let R_T^2 denote the proportion of the variance of y explained by all the $k + p$ covariates. The effect size is defined as $\delta = R_T^2 / (1 - R_T^2)$. The test statistic is then given by

$$F = \frac{\delta/\nu_1}{1/\nu_2} = \frac{R_T^2/\nu_1}{(1 - R_T^2)/\nu_2}$$

where $\nu_1 = k + p$ and $\nu_2 = n - k - p - 1$.

Testing a subset of coefficients: R^2 of full versus reduced models

Now suppose we are interested in testing whether the subset of p covariates accounts for any variation in the dependent variable after adjusting for the effects of the k covariates. The null hypothesis may be stated as $H_0: \beta_{k+1} = \dots = \beta_{k+p} = 0$. Let R_F^2 denote the R^2 of the full model with $k + p$ covariates and R_R^2 denote the R^2 of the reduced model with k covariates. With effect size $\delta = (R_F^2 - R_R^2)/(1 - R_F^2)$, the test statistic is given by

$$F = \frac{\delta/\nu_1}{1/\nu_2} = \frac{(R_F^2 - R_R^2)/\nu_1}{(1 - R_F^2)/\nu_2} \quad (4)$$

where $\nu_1 = p$ and $\nu_2 = n - k - p - 1$.

Testing a subset of coefficients: Partial multiple correlation

The F test for testing the null hypothesis $H_0: \beta_{k+1} = \dots = \beta_{k+p} = 0$ can also be constructed using partial multiple correlation. Let $\rho_p^2 = (R_F^2 - R_R^2)/(1 - R_R^2)$ denote the square of the partial multiple correlation. With effect size $\delta = \rho_p^2/(1 - \rho_p^2)$, the F statistic is given by

$$F = \frac{\delta/\nu_1}{1/\nu_2} = \frac{\rho_p^2/\nu_1}{(1 - \rho_p^2)/\nu_2}$$

where $\nu_1 = p$ and $\nu_2 = n - k - p - 1$. This is the same test statistic as in (4) for testing a subset of coefficients using R^2 .

Reference

Cohen, J. 1988. *Statistical Power Analysis for the Behavioral Sciences*. 2nd ed. Hillsdale, NJ: Erlbaum.

Also see

[PSS-2] **power** — Power and sample-size analysis for hypothesis tests

[PSS-2] **power oneslope** — Power analysis for a slope test in a simple linear regression

[PSS-2] **power pcorr** — Power analysis for a partial-correlation test in a multiple linear regression

[PSS-2] **power, graph** — Graph results from the power command

[PSS-2] **power, table** — Produce table of results from the power command

[PSS-5] **Glossary**

[R] **regress** — Linear regression

[R] **test** — Test linear hypotheses after estimation